Pensieve header: An implementation of the partial quadratic signature formalism for tangles; with Jessica Liu.

230109 **Def.** Given a v.s. V, a Partial Quadratic (PQ) Q on V is a symmetric bilinear form Q on a subspace $\mathcal{D}(Q) \subset V$. For $U \subset \mathcal{D}(Q)$, denote $\operatorname{ann}_Q(U) := \{v \in \mathcal{D}(Q) : Q(U, v) = 0\}$.
Def. $Q_1 + Q_2$ is with $\mathcal{D}(Q_1 + Q_2) = \mathcal{D}(Q_1) \cap \mathcal{D}(Q_2)$.
Def. Given a linear $\psi\colon V \to W$ and a PQ Q on W, the pullback is $(\psi^* Q)(v_1, v_2) = Q(\psi v_1, \psi v_2)$ with $\mathcal{D}(\psi^* Q) = \phi^{-1}(\mathcal{D}(Q))$.
Def. Given $\phi\colon V \to W$ and a PQ Q on V the pushforward $\phi_* Q$ is with $\mathcal{D}(\phi_* Q) = \phi(\operatorname{ann}_Q(\mathcal{D}(Q) \cap \ker\phi))$ and $(\phi_* Q)(w_1, w_2) = Q(v_1, v_2)$, where v_i are s.t. $\phi(v_i) = w_i$ and $Q(v_i, \operatorname{rad} Q|_{\ker\phi}) = 0$.

Thm(?). ψ^* and ϕ_* are well-defined and functorial, and if $\alpha /\!\!/ \beta = \gamma /\!\!/ \delta$, then $\gamma^* /\!\!/ \alpha_* = \delta_* /\!\!/ \beta^*$. ψ^* is additive but ϕ_* isn't.

$$\begin{tikzcd} \bullet \arrow[r, "\alpha"] \arrow[d, "\gamma"'] & \bullet \arrow[d, "\beta"] \\ \bullet \arrow[r, "\delta"'] \arrow[ur, dotted] & \bullet \end{tikzcd}$$

Thm(?). Over $\mathbb{R}$, given $\phi\colon V \to W$ and PQs Q on V and C on W,

$$\operatorname{sign}_V(Q + \phi^* C) = \operatorname{sign}_{\ker\phi}(\iota^* Q) + \operatorname{sign}_W(C + \phi_* Q).$$

For a knot K and a complex unit ω set $u = \Re(\omega^{1/2})$, $v = \Re(\omega)$, make an $F \times F$ matrix A with contributions







In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\People\\LiuJ"];
<< KnotTheory`
```

Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.

pdf

In[∘]:=
```
Kas[X[i_, j_, k_, l_]] := If[PositiveQ@X[i, j, k, l],
  Kas[Perm[{-i, j, k, -l}], PQ[Subspace[{y_{-i}, y_j, y_k, y_{-l}}, {y_{-i}, y_j, y_k, y_{-l}}],
    1/2 (η_{-i}^2 + 2 u η_{-i} η_j + v η_j^2 + 2 η_{-i} η_k + 2 u η_j η_k + η_k^2 + 2 u η_{-i} η_{-l} + 2 η_j η_{-l} + 2 u η_k η_{-l} + v η_{-l}^2)]],
  Kas[Perm[{-i, -j, k, l}], PQ[Subspace[{y_{-j}, y_k, y_l, y_{-i}}, {y_{-j}, y_k, y_l, y_{-i}}],
    1/2 (-v η_{-i}^2 - 2 u η_{-i} η_{-j} - η_{-j}^2 - 2 η_{-i} η_k - 2 u η_{-j} η_k - v η_k^2 - 2 u η_{-i} η_l - 2 η_{-j} η_l - 2 u η_k η_l - η_l^2)]]
 ]
```

pdf

In[∘]:=
```
CF[Subspace[{}, {0 ...}]] := Subspace[{}, {}];
CF[Subspace[vs_, {}]] := Subspace[Sort[vs], {}];
CF[Subspace[vs_, gens_]] := Module[{cvs = Sort[vs]},
  Subspace[cvs,
   DeleteCases[RowReduce[Table[Coefficient[g, v], {g, gens}, {v, cvs}]].cvs, 0]
  ]]
```

In[∘]:=
```
CF[Subspace[{y, z, x, w}, {x + y, x - y + z, x + 2 y + w}]]
```

Out[∘]=

$$\text{Subspace}\left[\{w, x, y, z\}, \left\{w + \frac{z}{2}, x + \frac{z}{2}, y - \frac{z}{2}\right\}\right]$$

pdf

In[∘]:=
```
Eval[Q_, v_, w_] := Expand[Q v w] //. {η_i_ y_i_ :> 1, η_i_^2 y_i_^2 :> 2} /. (η | y)_ → 0;
Eval[ϕ_, v_] := Expand[ϕ v] /. {η_i_ y_i_ :> 1, η_i_^2 y_i_ :> 2 η_i} /. y_ → 0;
```

In[∘]:=
```
Eval[u η1^2 + v η1 η2, y1 + y2]
```

Out[∘]=

$2\,u\,\eta_1 + v\,\eta_1 + v\,\eta_2$

In[∘]:=
```
Eval[Eval[u η1^2 + v η1 η2, y1], y1 + y2]
```

Out[∘]=

$2\,u + v$

In[∘]:=
```
Eval[u η1^2 + v η1 η2, y1 + y2, y1]
```

Out[∘]=

$2\,u + v$

pdf

In[∘]:=
```
Pivot[v_Plus] := v[[1]]; Pivot[v_] := v;
y_i_^* := η_i; η_i_^* := y_i; (vs_List)^* := Table[v^*, {v, vs}];
```

pdf

In[∘]:=
```
CF[PQ[sub_Subspace, Q_]] := Module[{csub, cvs, cgens},
  {cvs, cgens} = List @@ (csub = CF[sub]);
  PQ[csub, Sum[Eval[Q, v, w] Pivot[v]^* Pivot[w]^* / 2, {v, cgens}, {w, cgens}]]
 ]
```

In[∘]:=
```
CF[PQ[Subspace[{y1, y2, y3}, {y1 + 2 y3, y2 + 3 y3}], η3^2]]
```

Out[∘]=

PQ[Subspace[{y_1, y_2, y_3}, {$y_1 + 2\,y_3$, $y_2 + 3\,y_3$}], $4\,\eta_1^2 + 12\,\eta_1\,\eta_2 + 9\,\eta_2^2$]

In[∘]:=
```
Eval[η3^2, y1 + 2 y3, y2 + 3 y3]
```

Out[∘]=

12

In[∘]:=
```
Eval[4 η1^2 + 12 η1 η2 + 9 η2^2, y1 + 2 y3, y2 + 3 y3]
```

Out[∘]=

12

In[∘]:=
```
Eval[4 η1^2 + 12 η1 η2 + 9 η2^2, y1, y2]
```

Out[∘]=

12

In[∘]:=
```
Eval[12 η1 η2, y1, y2]
```

Out[∘]=

12

pdf

In[∘]:=
```
Perp[Subsp_] := Module[{pp, cvs, cgens},
  {cvs, cgens} = List @@ CF@Subsp;
  pp = Complement[cvs, Pivot /@ cgens]*;
  CF@Subspace[cvs*,
    Table[p - Sum[Coefficient[g, p*] Pivot[g]*, {g, cgens}], {p, pp}]
   ]
 ]
```

In[∘]:=
```
Perp@Subspace[{y1, y2, y3}, {y1 - y2}]
```

Out[∘]=

$\text{Subspace}[\{\eta_1, \eta_2, \eta_3\}, \{\eta_1 + \eta_2, \eta_3\}]$

In[∘]:=
```
Perp@Perp@Subspace[{y1, y2, y3}, {y1 - y2}]
```

Out[∘]=

$\text{Subspace}[\{y_1, y_2, y_3\}, \{y_1 - y_2\}]$

pdf

In[∘]:=
```
Id[vs_] := LT[vs, vs, Table[v → v, {v, vs}]]
```

In[∘]:=
```
Id[{y1, y2}]
```

Out[∘]=

$\text{LT}[\{y_1, y_2\}, \{y_1, y_2\}, \{y_1 \to y_1, y_2 \to y_2\}]$

pdf

In[∘]:=
```
LT[dom_, ran_, rs_]*[Subspace[ran_, gens_]] := Perp@CF@Subspace[dom*, Table[
     Sum[Eval[p, v /. rs] v*, {v, dom}],
     {p, Perp[Subspace[ran, gens]]⟦2⟧}
    ]]
```

In[∘]:=
```
LT[{y-1, y-2, y-3}, {y1, y2, y3}, {y-1 → y1 + 2 y3, y-2 → 2 y2 - y3, y-3 → y3}]*[
 Subspace[{y1, y2, y3}, {y1 - y2} ]]
```

Out[∘]=

$$\text{Subspace}\left[\{y_{-3}, y_{-2}, y_{-1}\}, \left\{y_{-3} + \frac{y_{-2}}{5} - \frac{2\, y_{-1}}{5}\right\}\right]$$

pdf

In[∘]:=
```
LT[dom_, ran_, rs_]⋆[Subspace[dom_, gens_]] := CF@Subspace[ran, gens /. rs]
```

In[∘]:=
```
LT[{y1, y2, y3}, {y1, y2}, {y1 → 0, y2 → y1, y3 → y2}]⋆[Subspace[{y1, y2, y3}, {y1, y3}]]
```

Out[∘]=

$\text{Subspace}[\{y_1, y_2\}, \{y_2\}]$

pdf

In[∘]:=
```
LT[dom_, ran_, rs_]*[PQ[sub_, Q_]] := CF@PQ[
   LT[dom, ran, rs]*[sub],
   Sum[Eval[Q, v1 /. rs, v2 /. rs] v1* v2* / 2, {v1, dom}, {v2, dom}]
  ]
```

In[∘]:= $\mathrm{Id}[\{y_1, y_2, y_3\}]^*[\mathrm{PQ}[\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3, y_2 + 3 y_3\}], 4 \eta_1^2 + 12 \eta_1 \eta_2 + 9 \eta_2^2]]$

Out[∘]=

$\mathrm{PQ}[\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3, y_2 + 3 y_3\}], 4 \eta_1^2 + 12 \eta_1 \eta_2 + 9 \eta_2^2]$

In[∘]:= $\mathrm{LT}[\{y_{-1}, y_{-2}, y_{-3}\}, \{y_1, y_2, y_3\}, \{y_{-1} \to y_1 + 2 y_3, y_{-2} \to 2 y_2 - y_3, y_{-3} \to y_3\}]^*[$
$\mathrm{PQ}[\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3, y_2 + 3 y_3\}], 4 \eta_1^2 + 12 \eta_1 \eta_2 + 9 \eta_2^2]]$

Out[∘]=

$\mathrm{PQ}\left[\mathrm{Subspace}\left[\{y_{-3}, y_{-2}, y_{-1}\}, \left\{y_{-3} + \frac{y_{-2}}{7}, y_{-1}\right\}\right], \frac{36 \eta_{-3}^2}{49} + \frac{24}{7} \eta_{-3} \eta_{-1} + 4 \eta_{-1}^2\right]$

In[∘]:= $\mathrm{Eval}\left[\frac{36 \eta_{-3}^2}{49} + \frac{24}{7} \eta_{-3} \eta_{-1} + 4 \eta_{-1}^2, y_{-3} + \frac{y_{-2}}{7}, y_{-3} + \frac{y_{-2}}{7}\right]$

Out[∘]=

$\frac{72}{49}$

In[∘]:= $\mathrm{Eval}\Big[4 \eta_1^2 + 12 \eta_1 \eta_2 + 9 \eta_2^2,$
$y_{-3} + \frac{y_{-2}}{7}$ /. $\{y_{-1} \to y_1 + 2 y_3, y_{-2} \to 2 y_2 - y_3, y_{-3} \to y_3\}$,
$y_{-3} + \frac{y_{-2}}{7}$ /. $\{y_{-1} \to y_1 + 2 y_3, y_{-2} \to 2 y_2 - y_3, y_{-3} \to y_3\}\Big]$

Out[∘]=

$\frac{72}{49}$

pdf

In[∘]:=
```
Subspace /: Subspace[vs_, gen1s_] + Subspace[vs_, gen2s_] :=
  CF@Subspace[vs, gen1s ⋃ gen2s]
Subspace /: sub1_Subspace ⋂ sub2_Subspace := Perp[Perp[sub1] + Perp[sub2]]
```

In[∘]:= $\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3\}] + \mathrm{Subspace}[\{y_1, y_2, y_3\}, \{3 y_3\}]$

Out[∘]=

$\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1, y_3\}]$

In[∘]:= $\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3\}] \bigcap \mathrm{Subspace}[\{y_1, y_2, y_3\}, \{2 y_3, y_1\}]$

Out[∘]=

$\mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3\}]$

pdf

In[∘]:=
```
Subspace /: v_ ∈ Subspace[vs_, gens_] :=
  (Subspace[vs, gens] ⋂ Subspace[vs, {v}])[[2]] =!= {}
```

In[∘]:= $y_3 \in \mathrm{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3\}]$

Out[∘]=

False

In[•]:=
```
y₃ ∈ Subspace[{y₁, y₂, y₃}, {y₁ + 2 y₃, y₁ + y₃}]
```
Out[•]=
```
True
```

pdf

In[•]:=
```
Ann_PQ[𝒟_Subspace,Q_][Subspace[vs_, gens_]] :=
  𝒟 ⋂ Perp@Subspace[vs*, Table[Eval[Q, g], {g, gens}]]
```

In[•]:=
```
Ann_PQ[Subspace[{y₁,y₂,y₃},{y₁+2 y₃,y₂+3 y₃}],4 η₁²+12 η₁ η₂+9 η₂²][Subspace[{y₁, y₂, y₃}, {y₁ + 2 y₃}]]
```
Out[•]=

$$\text{Subspace}\left[\{y_1, y_2, y_3\}, \left\{y_1 - \frac{2 y_2}{3}\right\}\right]$$

In[•]:=

$$y_1 - \frac{2 y_2}{3} \in \text{Subspace}[\{y_1, y_2, y_3\}, \{y_1 + 2 y_3, y_2 + 3 y_3\}]$$

Out[•]=
```
True
```

In[•]:=

$$\text{Eval}\left[4 \eta_1^2 + 12 \eta_1 \eta_2 + 9 \eta_2^2, y_1 - \frac{2 y_2}{3}, y_1 + 2 y_3\right]$$

Out[•]=
```
0
```

pdf

In[•]:=
```
Ker[LT[{}, _, _]] := Subspace[{}, {}];
Ker[LT[dom_, {}, _]] := Subspace[dom, dom];
Ker[LT[dom_, ran_, rs_]] := Module[{ns},
  ns = NullSpace[Table[Coefficient[d /. rs, r], {r, ran}, {d, dom}]];
  If[Length@ns > 0, CF@Subspace[dom, ns.dom], Subspace[dom, {}]]
 ]
```

In[•]:=
```
Ker[LT[{y₋₁, y₋₂, y₋₃}, {y₁, y₂, y₃}, {y₋₁ → y₁ + 2 y₃, y₋₂ → 2 y₂ - y₃, y₋₃ → y₃}]]
```
Out[•]=
```
Subspace[{y₋₁, y₋₂, y₋₃}, {}]
```

In[•]:=
```
Ker[LT[{y₋₁, y₋₂, y₋₃}, {y₁, y₂, y₃}, {y₋₁ → y₁ + 2 y₃, y₋₂ → -y₃, y₋₃ → y₃}]]
```
Out[•]=
```
Subspace[{y₋₃, y₋₂, y₋₁}, {y₋₃ + y₋₂}]
```

pdf

```
Section[LT[dom_, ran, rs_]]
Section[LT[Subspace[___], ran, rs_]]
```